
Excel -- Pivot Tables

Source : http://contextures.com/xlPivot01.html
Getting Started 
Many books and web sites have information on creating Pivot Tables in Excel 2003, and earlier versions. For a brief introduction, and list of Pivot Table links, visit Jon Peltier's site. 

You can download the zipped sample file used for this example.

Excel -- Pivot Tables -- Introduction

Use a Dynamic Data Source

You can use a dynamic formula to define the source range for a Pivot Table. As new items are added to the table, the named range will automatically expand. 

1. Name the Range
1. Choose Insert>Name>Define 

2. Type a name for the range, e.g. Database 

3. In the Refers To box, enter an Offset formula that defines the range size, based on the number of items in a column that doesn't contain any blank cells. , e.g.: 
=OFFSET(Data!$A$1,0,0,COUNTA(Data!$A:$A),7)
In this example, the list is on a sheet named 'Data', starting in cell A1. The arguments used in this Offset function are: 

1. Reference cell: Data!$A$1 

2. Rows to offset: 0 

3. Columns to offset: 0 

4. Number of Rows: COUNTA(Data!$A:$A) 

5. Number of Columns: 7
Note: for a dynamic number of columns, 
replace the 7 with: COUNTA(Data!$1:$1)
4. Click OK 

2. Base the Pivot Table on the Named Range 

Select a cell in the database 

2. Choose Data>PivotTable and PivotChart Report 

3. Select 'Microsoft Excel List or Database', click Next. 

4. For the range, type your range name, e.g. Database 

5. Click Next 

6. Click the Layout button 

7. Drag field buttons to the row, column and data areas 

8. Click OK, click Finish 

Excel -- Pivot Tables -- Data Field Layout

Arrange Multiple Data Fields


If you place two fields in the Data area of a Pivot Table, they appear vertically arranged. To make the data easier to read, you can rearrange the table.

2. Point to the grey button for the Data field

Hold the left mouse button, and drag the Data button onto the cell which contains the word 'Total' 

4. Release the mouse button 

The Data fields will now be arranged horizontally. 

Rename Data Fields

When you add fields to the Data area, they are renamed, e.g. 'Units' becomes 'Sum of Units'. There are several ways to change the names, but the following is probably the easiest.

1. Select the heading cell in the Pivot Table. 

2. Type a new heading. 

3. Press the Enter key. 

Note: The typed name can't be the same as the original field name, i.e., you can't change 'Sum of Units' to 'Units'. 

However, you can type the original field name, and add a space character at the end, e.g. 'Units '

Excel -- Pivot Tables -- Show and Hide Items

Show All Items -- Excel 2002

Each field button has an arrow that opens a dropdown list of items.

· In Excel 2002, the first checkbox, except in a Page Field, is Show All. 

· Add a check mark to this checkbox, to display all the items. 
Hide Items -- Excel 2002

You can hide all except one of the items in a field.

· Remove the check mark from the Show All checkbox, to remove all the check marks in the list 

· Check at least one of the items, then click OK.

Show All Items -- Excel 97/Excel 2000

In previous versions of Excel, where there is no Show All checkbox, you can use programming to show or hide multiple fields. The following code will show all items in all row fields. 

To show column fields, change pt.RowFields to pt.ColumnFields. 


To show all fields, change pt.RowFields to pt.VisibleFields.

Sub PivotShowItemAllVisible()

'sort is set to Manual to prevent errors, e.g.

'unable to set Visible Property of PivotItem class
Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

For Each pt In ActiveSheet.PivotTables

   For Each pf In pt.RowFields

    pf.AutoSort xlManual, pf.SourceName

     For Each pi In pf.PivotItems

         pi.Visible = True

     Next pi

    pf.AutoSort xlAscending, pf.SourceName

   Next pf

Next pt

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub       
Hide Items -- Excel 97/Excel 2000

In previous versions of Excel, where there is no Show All checkbox, you can use programming to show or hide multiple fields. The following code hides all items, except the last item, in all row fields. 

To hide column fields, change pt.RowFields to pt.ColumnFields.

Show Items in Specific Field -- Excel 97/Excel 2000

The following code will prompt you for a field name, and will show all items in the specified field.

Sub PivotShowItemsField()

Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

Dim strPF As String

Set pt = ActiveSheet.PivotTables(1)

strPF = InputBox("What Field?", "Field Name")

Set pf = pt.PivotFields(strPF)

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

   With pf

    .AutoSort xlManual, .SourceName

     For Each pi In .PivotItems

         pi.Visible = True

     Next pi

    .AutoSort xlAscending, .SourceName

    End With

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub       
Hide Items in Specific Field -- Excel 97/Excel 2000

The following code will prompt you for a field name, and will hide all items in the specified field.

Sub PivtoHideItemsField()

Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

Dim strPF As String

Set pt = ActiveSheet.PivotTables(1)

strPF = InputBox("What Field?", "Field Name")

Set pf = pt.PivotFields(strPF)

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

   With pf

    .AutoSort xlManual, .SourceName

     For Each pi In pf.PivotItems

         pi.Visible = False

     Next pi

    .AutoSort xlAscending, .SourceName

    End With

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub

Show Specific Item in Specific Field 
The following code will prompt you for a field name and item name, and will hide all other items in the field.

Sub PivotShowSpecificItems()

Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

Dim strPromptPF As String

Dim strPromptPI As String

Dim strPF As String

Dim strPI As String

strPromptPF = "Please enter the name of the field you wish to filter."

strPromptPI = "Please enter the item you wish to filter for."

Set pt = ActiveSheet.PivotTables(1)

strPF = InputBox(strPromptPF, "Enter Field Name")

strPI = InputBox(strPromptPI, "Enter Item")

Set pf = pt.PivotFields(strPF)

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

With pf

    .AutoSort xlManual, .SourceName

    For Each pi In pf.PivotItems

        pi.Visible = False

    Next pi

    .PivotItems(strPI).Visible = True

    .AutoSort xlAscending, .SourceName

End With

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub       

Hide Page Field Items

A Page Field dropdown doesn't have checkboxes that allow you to select multiple items. You can select All, or an individual item to display. 

Use the following technique to display multiple items from a Page field.

· Double-click the Page field button, to display the PivotTable Field dialog box. 

· [image: image11.png]Hide ffems:



From the Hide items list, select the items you want to hide. 

· Click OK. 

Or

· Drag the Page field button to the Row area. 

· Use the check boxes to select or deselect items 

· Drag the Page field button back to the Page area 

The Page Field displays Multiple Items, and the Pivot table only shows results for the Page items that are not hidden.

Show Specific PivotChart Items

The following code will prompt you for a field name, and will show all items in the specified field

Sub PivotChartShowItemsField()

Dim ch As Chart

Dim pf As PivotField

Dim pi As PivotItem

Dim strPF As String

Set ch = ActiveChart

strPF = InputBox("What Field?", "Field Name")

Set pf = ch.PivotLayout.PivotFields(strPF)

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

    pf.AutoSort xlManual, pf.SourceName

     For Each pi In pf.PivotItems

         pi.Visible = True

     Next pi

    pf.AutoSort xlAscending, pf.SourceName

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub       
Hide Specific PivotChart Items

The following code will prompt you for a field name, and will hide all items in the specified field.

Sub PivotChartHideItemsField()

Dim ch As Chart

Dim pf As PivotField

Dim pi As PivotItem

Dim strPF As String

Set ch = ActiveChart

strPF = InputBox("What Field?", "Field Name")

Set pf = ch.PivotLayout.PivotFields(strPF)

Application.ScreenUpdating = False

Application.DisplayAlerts = False

On Error Resume Next

   With pf

    .AutoSort xlManual, .SourceName

     For Each pi In pf.PivotItems

         pi.Visible = False

     Next pi

    .AutoSort xlAscending, .SourceName

    End With

Application.DisplayAlerts = True

Application.ScreenUpdating = True

End Sub

Hide Pivot Subtotals

When you add more than one field to the row or column area, the pivot table will display subtotals. The following code removes all subtotals from the pivot table. 

Sub NoSubtotals()

'turns off subtotals in pivot table

'.PivotFields could be changed to

'.RowFields or .ColumnFields
Dim pt As PivotTable

Dim pf As PivotField

On Error Resume Next

For Each pt In ActiveSheet.PivotTables

  For Each pf In pt.PivotFields

    'First, set index 1 (Automatic) to True,

    'so all other values are set to False

    pf.Subtotals(1) = True

    pf.Subtotals(1) = False

  Next pf

Next pt

End Sub

Excel -- Pivot Tables -- Clear Old Items
Old Items Remain in Pivot Field Dropdowns

The data in the pivot table source may change, and items that were previously in the source are no longer included. For example, some sales reps may leave the company, and the names of their replacements appear in the source table.

Even after you refresh the pivot table, the names of the old sales reps will appear, along with the new names. In the list at right, Cartier has replace Gill, but Gill still appears in the list.

Manually Clear Old Items

To manually clear the old items from the list:

1. If you manually created any groups that include the old items, ungroup those items. 

2. Drag the pivot field out of the pivot table. 

3. On the Pivot toolbar, click the Refresh button 

4. Drag the pivot field back to the pivot table 
Programmatically Clear Old Items -- Excel 2002 or later

In Excel 2002, and later versions, you can programmatically change the pivot table properties, to prevent missing items from appearing, or clear items that have appeared. 
Sub DeleteMissingItems2002All()

'prevents unused items in non-OLAP PivotTables

Dim pt As PivotTable

Dim ws As Worksheet

Dim pc As PivotCache

'change the settings
For Each ws In ActiveWorkbook.Worksheets

  For Each pt In ws.PivotTables

    pt.PivotCache.MissingItemsLimit = xlMissingItemsNone

  Next pt

Next ws

'refresh all the pivot caches
For Each pc In ActiveWorkbook.PivotCaches

  On Error Resume Next
  pc.Refresh

Next pc

End Sub 

Programmatically Clear Old Items -- Excel 97/Excel 2000

In previous versions of Excel, run the following code to clear the old items from the dropdown list.

Sub DeleteOldItemsWB()

'gets rid of unused items in PivotTable

' based on MSKB (202232)
Dim ws As Worksheet

Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

On Error Resume Next
For Each ws In ActiveWorkbook.Worksheets

  For Each pt In ws.PivotTables

    pt.RefreshTable

    pt.ManualUpdate = True
    For Each pf In pt.VisibleFields

      If pf.Name <> "Data" Then

        For Each pi In pf.PivotItems

          If pi.RecordCount = 0 And _

            Not pi.IsCalculated Then
            pi.Delete

          End If
        Next pi

      End If
    Next pf

    pt.ManualUpdate = False
    pt.RefreshTable

  Next pt

Next ws

End Sub       
Excel -- Pivot Tables -- Field Settings

Manually Hide or Show Subtotals

To manually hide subtotals for a field:

1. Double-click the field button, to open the PivotTable field dialog box. 

2. For Subtotals, select 'None' 

3. Click OK 
To manually show subtotals for a field:

1. Double-click the field button, to open the PivotTable field dialog box. 

2. For Subtotals, select 'Custom' 

3. Select one of the functions from the list, e.g. 'Average' 

4. Click OK

Programmatically Hide Subtotals

You can use a macro to hide subtotals in a PivotTable. This example uses the PivotField property, to hide all the subtotals.
To hide only the Row field subtotals, use the RowFields property. 
To hide only the Column field subtotals, use the ColumnFields property. 
Sub NoSubtotals()

'turns off subtotals in pivot table

'.PivotFields could be changed to

'.RowFields or .ColumnFields

Dim pt As PivotTable

Dim pf As PivotField

For Each pt In ActiveSheet.PivotTables

  For Each pf In pt.PivotFields

    'Set index 1 (Automatic) to True,

    'so all other values are set to False

    pf.Subtotals(1) = True
    pf.Subtotals(1) = False
  Next pf

Next pt

End Sub   

Show Items with No Data

By default, the Pivot Table shows only the items for which there is data. In the example shown at right, not all colours were sold each day. You may wish to see all the items each day, even those with no data.

1. Double-click the field button, to open the PivotTable field dialog box. 

2. Check the 'Show items with no data' check box. 

3. Click OK 

Show Top Items Only

Instead of showing all the items in a field, you can restrict the Pivot Table to show only the top (or bottom) items.

1. Double-click the field button, to open the PivotTable field dialog box. 

2. Click the Advanced button 

3. Under 'Top 10 AutoShow', select On. 

4. For 'Show', select Top or Bottom 

5. Click the Scroll buttons, or type, to enter the number of items to show. 

6. Click OK, click OK 
Sort All Fields in Ascending Order

After adding new records to your data, new items may appear at the end of the existing data, instead of being listed alphabetically. The following code will sort all fields in all pivot tables.

Sub SortAllFields()

On Error Resume Next

Application.ScreenUpdating = False

Dim pt As PivotTable
Dim ws As Worksheet
Dim pf As PivotField
For Each ws In ActiveWorkbook.Worksheets
  For Each pt In ws.PivotTables
    pt.ManualUpdate = True

    For Each pf In pt.PivotFields

        pf.AutoSort xlAscending, pf.Name
    Next pf
    pt.ManualUpdate = False

  Next pt
Next ws
Application.ScreenUpdating = True

End Sub

Excel -- Pivot Tables -- GetPivotData

Generate GetPivotData

In Excel 2002, and later versions, when you type an equal sign, then click on a cell in a pivot table, you may see a GETPIVOTDATA formula, instead of a cell reference. 

If you prefer to use a cell reference, you can:

· type the reference, e.g. =$B$5 

or 

· add the Generate GetPivotData button to a toolbar, to turn this feature off.

To add a Generate GetPivotData button: 

On the PivotTable toolbar, click the Toolbar Options button 

2. Click the Add or Remove Buttons command 

3. Click PivotTable, to open the submenu 

4. Near the end of the commands list, click on Generate GetPivotData to add a check mark. 

5. Click on the worksheet, to close the menu.

On the PivotTable toolbar, click the Generate GetPivotData button to toggle this feature on and off.

[image: image21.png]A B [ D

1 [Rep [
2]

3] S«mufunﬁ

4 [Region _ wTotal

5 |Abeta |

6 |Ontario

7 |Quebec

8 [Grand Total 2554]
9

10

1

[SGETPIVOTDATA("Units" $A53,"Region”.




GetPivotData Formula 

To extract data from a PivotTable, you can use the GetPivotData function. The function arguments depend on the version of Excel that you're using, so check Excel's on-line help for examples. 

The example at the right is for Excel 2002, and returns the number of Units sold in Ontario. In Excel 2002, if you have the Generate GetPivotData feature turned on, this formula will be created automatically, when you reference a cell in the Pivot Table.

In Excel 2000 or Excel 97, you have to create the GetPivotData formula manually, by typing an equal sign, the function name, and the required arguments, just as you would for any other function. 

This example shows the function arguments required for Excel 2000  or Excel 97, to return the number of Units sold in Ontario.

This example shows the function arguments required for Excel 2000 or Excel 97, to return the total number of Units sold. 

Using Cell References in GetPivotData

Instead of typing item or field names in the GetPivotD  ata arguments, you can refer to worksheet cells. In this example for Excel 2002, cell A10 contains one of the arguments, and the formula refers to that cell.

In this example for Excel 2000, cell A10 contains one of the arguments, and the formula refers to that cell. A space is included at the end of "Units ". If this is omitted, the formula will return an #N/A error.

[image: image26.png]10

Ontario_JUnits

1

EGETPIVOTDATA(SAS3.510 & " " &A10)




In the next example, both arguments come from cell  references, and a space character is concatenated in the middle, using the & operator. 
Excel -- Pivot Tables -- Grouping Data

In a Pivot Table, you can group the items in a Row or Column  field. 

For example, items in a date field can be grouped by month, and items in a number field can be grouped by tens.

Grouping Dates

To group the items in a Date field

1. Right-click the Date field button. 

2. Choose Group and Show Detail | Group 

3. In the Grouping dialog box, select one or more options from the 'By' list. 

To limit the dates that are grouped, you can set a Start and End date, by typing the dates in the 'Starting at' and 'Ending at' boxes 

5. Click OK 

Grouping Dates by Week

To group the items in a Date field by week
1. Right-click the Date field button. 

2. Choose Group and Show Detail | Group 

3. In the Grouping dialog box, select Days from the 'By' list. 

4. For 'Number of days', select 7 

5. The week range is determined by the date in the 'Starting at' box, so adjust this if necessary. In the example at right, December 29, 2003 (a Monday), was entered as the starting date. 

6. Click OK 
Problems when Grouping Data 

If you try to group a date or field, you may see an error message that says, "Cannot group that selection."

This problem usually occurs when the field contains records with a blank date/number field, or text in a date/number field. To fix the problem

For blank cells, fill in the date/number (use a dummy date/n umber if necessary). 

· If there is text in the date/number field, remove it. 

· If numbers are being recognized as text change them. 

If you don't have blank cells or text in the date column, there may be a grouped field left over from the previous time that you grouped the data.

1. Check the field list, to see if there's a second copy of the date field, e.g. Date2. 

2. If there is, add it to the row area, and ungroup it. 

3. Then, you should be able to group the date field again

Count Unique Items

In a pivot table, you may want to know how many unique customers placed an order for an item, instead of how many orders were placed. A pivot table won't calculate a unique count. However, you could add a column to the database, then add that field to the pivottable. 

For example, to count the unique occurences of a Customer/Item order, add a column to your database, with the heading 'CustItem' 

In the first data row, enter a formula that refers to the customer and item columns. For example:

=IF(SUMPRODUCT(($A$2:$A2=A2)*($C$2:$C2=C2))>1,0,1)

Copy the formula down to all rows in the database.

Then, add the field to the data area of the pivot table.

In this example, you can see that nine unique customers placed an order for binders, and there were 14 orders for binders.

Excel -- Pivot Tables -- Multiple Consolidation Ranges

To create a Pivot Table, you can use data from different sheets in a workb ook, or from different workbooks, if those tables have identical column structures. 

However, you won't get the same pivot table layout that you'd get from a single range. 

Source data 
Create a pivot table from multiple consolidation ranges

Choose Data | PivotTable and PivotChart Report 

2. Select Multiple consolidation ranges, click Next 

3. Select one of the page options, click Next 

4. Select each range, and click Add 

5. If you chose 'I will create the page fields', you can select each range, and assign field names, in step 2b 

Click Next

5.  Select a location for the PivotTable, click Finish 

6. In the Column dropdown, hide any columns that contain meaningless data. 

Limitations of Multiple Consolidation Ranges

In this example, Item is the first column in the data source, and the pivot table row heading shows the item names. Remaining fields are shown in the column area.

You can change the function (e.g. SUM) that's being used by the data value, but it will use the same function on all these columns. The Pivot Table contains some meaningless data, such as sum of Date and columns full of zeros where the database columns contain text. 

To avoid this, you can rearrange your database columns, and then use data ranges that only include the columns that you want to total. 

If possible, move your data to a single worksheet, or store it in a database, such as MS Access, and you'll have more flexibility in creating the pivot table.

Or, you can create named ranges in an Excel file, and use MS Query to combine the data. There are sample files here: http://www.contextures.com/excelfiles.html#PT0007
Excel -- Pivot Tables -- Printing
Check for Pivot Items Before Printing

The following code loops through a list of employee names on the Lists worksheet. It will test for each name in the Employee page field, before changing the current page. If the employee name is an item in the field, the page field will be changed, and the pivot table will be printed. Use the Preview:=True setting for testing. When ready to print, change to Preview:=False 
Sub ChangePivotPage()

'Test if Item exists

'before setting the CurrentPage to that item
Dim ws As Worksheet

Dim pt As PivotTable

Dim pi As PivotItem

Dim str As String

Dim rng As Range

Dim c As Range

Set ws = ActiveSheet

Set pt = ws.PivotTables(1)

Set rng = Worksheets("Lists").Range("EmpNames")

For Each c In rng

  Set pi = Nothing

  str = c.Value

  With pt.PageFields("Employee")

    On Error Resume Next

    Set pi = .PivotItems(str)

    On Error GoTo 0

    If pi Is Nothing Then

      Debug.Print str & " was NOT printed"

    Else

      .CurrentPage = str

      ws.PrintOut Preview:=True

    End If

  End With

Next c

End Sub     
Print Pivot Table for each Page Item

The following code will print the pivot table once for each item in the page field (assumes there is one page field). Use the PrintPreview line for testing. When ready to print, remove the apostrophe from the beginning of the ActiveSheet.PrintOut line, and add an apostrophe to the beginning of the ActiveSheet.PrintPreview line. 
Sub PrintPivotPages()

 'prints a copy of pivot table for each item in page field

 'assumes one page field exists

On Error Resume Next

Dim pt As PivotTable

Dim pf As PivotField

Dim pi As PivotItem

Set pt = ActiveSheet.PivotTables.Item(1)

  For Each pf In pt.PageFields

    For Each pi In pf.PivotItems

      pt.PivotFields(pf.Name).CurrentPage = pi.Name

'      ActiveSheet.PrintOut  'use this for printing

      ActiveSheet.PrintPreview  'use this for testing

    Next

  Next pf

End Sub     
Print Pivot Table for Each Page Item - Multiple Page Fields

The following code will print the pivot table for each combination of page items. If PrintFlag is not set to true, descriptive information is written to the PageItemList worksheet.

Option Compare Text

Public mrow As Integer

Public PrintFlag As Boolean

'==========================================

Sub PrintAllPages()

'from code posted by Tom Ogilvy

'September 5 2004
Dim holdSettings

Dim ws As Worksheet

Dim wsPT As Worksheet

Set ws = Worksheets("PageItemList") 'sheet for page items

Set wsPT = Worksheets("Pivot") 'sheet with PivotTable

mrow = 0

If MsgBox("Print?", vbYesNo, "Print?") = vbYes Then

  PrintFlag = True

Else

  PrintFlag = False

  MsgBox "Page field items will be listed on sheet " & ws.Name

End If

If Not PrintFlag Then

  ws.Cells(1, 1).CurrentRegion.Clear

End If

Set PvtTbl = wsPT.PivotTables(1)

wsPT.Activate

If PvtTbl.PageFields.Count = 0 Then

  MsgBox "The PivotTable has no Pages"

  Exit Sub

End If

With PvtTbl

ReDim holdSettings(1 To .PageFields.Count)

I = 1

For Each PgeField In .PageFields

  holdSettings(I) = PgeField.CurrentPage.Name

  I = I + 1

  PgeField.CurrentPage = PgeField.PivotItems(1).Name

Next PgeField

End With

PvtPage = 1

PvtItem = 1

DrillPvt oTable:=PvtTbl, Ipage:=PvtPage, wksht:=ws

I = 1

For Each PgeField In PvtTbl.PageFields

 PgeField.CurrentPage = holdSettings(I)

  I = I + 1

Next PgeField

End Sub

'===========================================

Sub DrillPvt(oTable, Ipage, wksht)

'Debug.Print "in DrillPvt, page:=" & Ipage & " Page Item: " & _

'  oTable.PageFields(Ipage).CurrentPage & " " & mrow

If Ipage = oTable.PageFields.Count Then

 With oTable

  For I = 1 To .PageFields(Ipage).PivotItems.Count

   .PageFields(Ipage).CurrentPage = _

   .PageFields(Ipage).PivotItems(I).Name

   mrow = mrow + 1

   slist = ""

   For j = 1 To .PageFields.Count

     slist = slist & .PageFields(j).CurrentPage & " "

   Next j

 '  Debug.Print slist

   If PrintFlag Then

''    ActiveSheet.PrintOut  'print the sheet

    ActiveSheet.PrintPreview  'preview -- for testing

   Else

    For j = 1 To .PageFields.Count

     wksht.Cells(mrow, j).Value = _

      .PageFields(j).CurrentPage.Name

    Next j

   End If

  Next I

 End With

 For I = oTable.PageFields.Count - 1 To 1 Step -1

   For j = 1 To oTable.PageFields(I).PivotItems.Count

     If oTable.PageFields(I).CurrentPage = _

      oTable.PageFields(I).PivotItems(j).Name Then

        CurrItem = j

        Exit For

     End If

   Next j

   If CurrItem <> oTable.PageFields(I).PivotItems.Count Then

      oTable.PageFields(I).CurrentPage = _

        oTable.PageFields(I).PivotItems(CurrItem + 1).Name

      Ipage = I + 1

      DrillPvt oTable, Ipage, wksht

   Else

     If I <> 1 Then

       oTable.PageFields(I).CurrentPage = _

        oTable.PageFields(I).PivotItems(1).Name

     Else

       Exit Sub

     End If

   End If

 Next I

Else

 DrillPvt oTable, Ipage + 1, wksht

End If

End Sub       
Excel -- Pivot Tables -- Custom Calculations

In a Pivot Table, you can summarize the data by using the values in other cells in the data area. 

For example, you can show each Region's total as a percentage of the national total. Or, calculate the difference between the sales totals for the current year, and the sales totals for the previous year.

% of Column

In this example, the pivot table has Region in the Row area, and Total in the Data area. A custom calculation will be added, to show the percentage for each region's sales, compared to the national total.

1. From the Pivot Table field list, drag another copy of the Total field to the Data area. 

2. If the data fields are arranged vertically, you can change them to a horizontal layout. 

3. Right-click the heading cell for the new column, and select Field Settings... 

4. In the Field Settings dialog box, type a name for the field, e.g. %Sales 

5. Click the Options button, to e  xpand the dialog box 

6. From the Show data as dropdown list, select % of column 

7. Click the OK button 
% of Row
[image: image42.png]Sum of Total [Regior |

item ~|Alberta_ Ontario Quebec |Grand Tosl|
Binder 7% 29%  13%|  49%|
Desk 4% 4% 0% 9%
Pen 1% 3% 7%|  10%
Pen Set 0% 12% 9%  21%|
Pencil 1% 8% 2% 1%
[Grand Toa! 13%  57%  31%] _ 100%)




In this example, the pivot table has Item in the Row area, Region in the Column area, and Total in the Data area. The total will be changed to a custom calculation, to show the percentage for e  ach region's sales of an item, compared to the item total.

1. Right-click one of the cells in the Data area, and select Field Settings... 

2. In the Field Settings dialog box, type a name for the field, e.g. %Sales 

3. Click the Options button, to expand the dialog box 

4. From the Show data as dropdown list, select % of row 

5. Click the OK button 
% of Total 

In this example, the pivot table has Item in the Row area, Region in the Co  lumn area, and Total in the Data area. 

The total will be changed to a custom calculation, to show the percentage for each region's sales of an item, compared to the Sales Grand Total for all Items.

1. [image: image43.png]


Right-click one of the cells in the Data area, and select Field Settings... 

2. In the Field Settings dialog box, type a name for the field, e.g. %Sales 

3. Click the Options button, to expand the dialog box 

4. From the Show data as dropdown list, select % of Total 

5. Click the OK button 

Difference from 

In this example, the pivot table has Region in the Row area, and Total in the Data area. Date is in the Column area, grouped by Year

The total will be changed to a custom calculation, to compare the current year's sales for each region, to previous year's sales, in dollars.

1. Right-click one of the cells in the Data area, and select Field Settings... 

In the Field Settings dialog box, type a name f  or the field, e.g. Change 

3. Click the Options button, to expand the dialog box 

4. From the Show data as dropdown list, select Difference From 

5. From the Base field list, choose Years 

6. From the Base item list, choose (previous) 

7. Click the OK button 
% Difference from 

In this example, the pivot table has Item in the Row area, and Total in the Data area. Date is in the Column area, grouped by Year

The total will be changed to a custom calculation, to compare the current year's sales for each Item, to previous year's sales, as a percentage. 

1. Right-click one of the cells in the Data area, and select Field Settings... 

2. In the Field Settings dialog box, type a name for the field, e.g. %Change 

3. Click the Options button, to expand the dialog box 

4. From the Show data as dropdown list, select % Difference From 

5. From the Base field list, choose Years 

6. From the Base item list, choose (previous) 

7. Click the OK button 

% Of 

In this example, the pivot table has Item in the Row area, Region in  the Column area, and Total in the Data area.

The total will be changed to a custom calculation, to compare each Region's sales to Ontario's sales, as a percentage.

1. Right-click one of the cells in the Data area, and select Field Settings... 

2. In the Field Settings dialog box, type a name for the field, e.g. %Ontario 

3. Click the Options button, to expand the dialog box 

From the Show data as dropdown list, select % Of 

5. From the Base field list, choose Region 

6. From the Base item list, choose Ontario 

7. Click the OK button

Running Total in


In this example, the pivot table has Region in the Column area, and Total in the Data area. Date is in the Row area, grouped by Year and Quarter

The total will be changed to a custom calculation, to calculate a running total of sales for each Region, over each Year.

1. Right-click one of the cells in the Data area, and select Field Settings... 

2. In the Field Settings dialog box, type a name for the field, e.g. Sales 

3. Click the Options button, to expand the dialog box 

4. From the Show data as dropdown list, select Running Total in 
5. From the Base field list, choose Date 

6. Click the OK button 

Remove a Custom Calculation 

To remove a custom calculation from a pivot table:

1. Click the dropdown arrow on the Data field button 

2. Remove the check mark from the custom calculation. 

Note: If you remove an item from the data area, it's removed from the pivot table. To replace it, you can drag it back from the pivot table field list.

Excel -- Pivot Tables -- Pivot Cache
Show the Pivot Table's CacheIndex

You can display a pivot table's CacheIndex number by using the following User Defined Function. Store the function code in a worksheet module. Then, on the worksheet, enter the formula:

=ShowCacheIndex(A3)

replacing A3 with a cell in your pivot table.

Function ShowCacheIndex(rngPT As Range) As Long

  ShowCacheIndex = rngPT.PivotTable.CacheIndex

End Function       
Show the Pivot Cache Memory Used

You can display the memory used by a pivot cache, by using the following User Defined Function. Store the function code in a worksheet module. Then, on the worksheet, enter the formula:

=GetMemory(A3)/1000

replacing A3 with a cell in your pivot table. The result is displayed in kilobytes.

Function GetMemory(rngPT As Range) As Long

 Dim pt As PivotTable

 Set pt = rngPT.PivotTable

 GetMemory = ActiveWorkbook _

 .PivotCaches(pt.CacheIndex).MemoryUsed

End Function 
Show the Pivot Cache Record Count

You can display the number of records in a pivot cache, by using the following User Defined Function. Store the function code in a worksheet module. Then, on the worksheet, enter the formula:

=GetRecords(A3)

replacing A3 with a cell in your pivot table.

Function GetRecords(rngPT As Range) As Long

 Dim pt As PivotTable

 Set pt = rngPT.PivotTable

 GetRecords = ActiveWorkbook _

 .PivotCaches(pt.CacheIndex).RecordCount 

End Function 

Change the Pivot Cache

If you have created several pivot tables in a workbook, you may find it more efficient to use the same pivot cache for all the pivot tables. The following code will change the pivot cache for each pivot table in the workbook.

Sub ChangePivotCache()

 'change pivot cache for all pivot tables in workbook

 Dim pt As PivotTable

 Dim wks As Worksheet

 For Each wks In ActiveWorkbook.Worksheets

 For Each pt In wks.PivotTables

 
pt.CacheIndex = Sheets("Pivot").PivotTables(1).CacheIndex

 Next pt

 Next wks 

End Sub 
Excel -- Pivot Tables -- Protection

Disable Selection 
In Excel 2002, and later versions, you can use a macro to disable selection in a pivot table. The dropdown arrows will disappear from the field buttons, and you can't change the displayed items.
Note: Instead of .PivotFields, you can use .RowFields, .ColumnFields or .PageFields
Sub DisableSelection()

Dim pt As PivotTable 
Dim pf As PivotField

Set pt = ActiveSheet.PivotTables(1)

  For Each pf In pt.PivotFields

      pf.EnableItemSelection = False

  Next pf

End Sub

Enable Selection 
The following macro enables the features that were turned off in the DisableSelection macro, shown above. (Excel 2002 and later versions.) 

Note: Instead of .PivotFields, you can use .RowFields, .ColumnFields or .PageFields
Sub EnableSelection()

Dim pt As PivotTable

Dim pf As PivotField

Set pt = ActiveSheet.PivotTables(1)

  For Each pf In pt.PivotFields

      pf.EnableItemSelection = True

  Next pf

End Sub 
Restrict PivotTable 

You can use a macro to disable many of the features in a pivot table. The following code prevents users from dragging any of the fields to a different location, or off the pivot table. It also prevents them from using the Pivot Wizard, Field List, Field Settings, or Drilldown, and disables the Refresh feature.
Note: Instead of .PivotFields, you can use .RowFields, .ColumnFields or .PageFields
Sub RestrictPivotTable()

Dim pf As PivotField

With ActiveSheet.PivotTables(1)

  .EnableWizard = False

  .EnableDrilldown = False

  .EnableFieldList = False   'Excel 2002+

  .EnableFieldDialog = False

  .PivotCache.EnableRefresh = False

  For Each pf In .PivotFields 

    With pf

      .DragToPage = False

      .DragToRow = False

      .DragToColumn = False

      .DragToData = False

      .DragToHide = False

    End With

  Next pf

End With

End Sub   
Allow PivotTable 

The following macro enables the features that were turned off in the RestrictPivotTable macro, shown above.

Note: Instead of .PivotFields, you can use .RowFields, .ColumnFields or .PageFields
Sub AllowPivotTable()

Dim pt As PivotTable

Dim pf As PivotField

Set pt = ActiveSheet.PivotTables(1)

With pt

  .EnableWizard = True

  .EnableDrilldown = True

  .EnableFieldList = True   'Excel 2002+

  .EnableFieldDialog = True

  .PivotCache.EnableRefresh = True

  For Each pf In pt.PivotFields

    .DragToPage = True 

    .DragToRow = True

    .DragToColumn = True

    .DragToData = True

    .DragToHide = True

  Next pf

End With

End Sub
Excel -- Pivot Tables -- Grand Totals
[image: image56.png]PivotTable Field Layout:

Display Options.
 Show items i tabular form
 Show ftems it odtine form!

I Display subtotals at top of aroup.
I~ ert blank line after each tem

Print Option

I Insert page break after eachitem

[ | ==

I~ Showitems with no data




Show Grand Total at the Top of a Pivot Table

There's no setting that allows you to display the grand total at the top of a pivot table. However, with this technique you can use another field that acts as the grand total,  and display it at the top.

Note: The instructions are for Excel 2003 and earlier.

Create the fake "Grand Total" field

1. In the source data, add a column with the heading GT, or use a space character as the column heading 

2. In every row of the source data, for the GT field, enter: Grand Total 

3. Add this field to the pivot table, as the first field in the Row area 

Change the Field Settings

1. In the pivot table, right-click on the GT field button, and click Field Settings 

2. Click the Layout button 

3. Click Show Items in Outline Form 

4. Add a check mark to Display Subtotals at Top of Group  

5. Click OK, twice. 

Hide the original Grand Total 

1. Right-click on a cell in the Pivot Table 

2. Click Table Options 

3. Remove the check mark for Grand Totals for Columns 

4. Click OK

1

